Enumerate reaction networks composed with heteromultimer and single transformations
November 2014

This particular code is to enumerate all possible reaction networks (with certain dimensions, $m \times n$ where n is speciese number m is reaction number) composed of three types of interactions:

$$A + B \rightarrow C$$
$$\mathrm{D} \rightarrow E + F$$
$$G \rightarrow H$$

Those are heterodimerization, disassociation and single transformation. However, we did not include $I \rightarrow 2J$ and $2K \rightarrow L$ those two type of elementary reaction here, we should implement those in near future.

With three types of elementary reactions above, we could construct a set of reaction networks, then we could use DSR graphs and/or bipartie graph to characteristic those networks whether they are multistationary and has closed competition loop (as well as interchangeable competitors).

The main purpose of this document is to explain the procedures that how to construct and enumerate all possible reaction networks when given fixed reaction number and species number.

If a chemical reaction network has m reactions driven by n chemical species, we could have a stoichiometric matrix $N_{m\times n}$ with dimension

But before to go through such checking, we need to preclude situations that clearly not a complex balanced reaction network, by which mean it obeys the following three constraints:

0. Only allow elementary reactions described above, which is the starting point to construct the matrix; (NB: we don't consider birth-death process like $\varnothing \rightarrow X$ and $Y \rightarrow \varnothing$)

a). list all possibly reaction vectors N_i $(i = 1..n)$ and select m of those into matrix (sequence does not matter)

The total number of reaction vectors for $G \rightarrow H$ is $\binom{n}{2}\cdot 2$ (since we need to consider the sides of two species in a reaction);

The total number of reaction vectors for $A + B \rightarrow C$ and $\mathrm{D} \rightarrow E + F$ are both $\binom{n}{3}\cdot 3$ (same here);

We get $r = \binom{n}{2}\cdot 2 + \binom{n}{3}\cdot 3 + \binom{n}{3}\cdot 3$ number of reaction vectors, we store it in matrix $R_{r\times n}$, then we construct all the stoichiometric matrix by choosing m reaction vectors from $R_{r\times n}$ into $N_{m\times n}$, therefore we have total number of $\binom{r}{m}$ matrices to construct. Each constructed matrix $N_{m\times n}$ will go through balanced checking, mass conservation checking to become a valid stoichiometric matrix to go through further bistability check and competition check.

b). (Optional) We could further reduce the number of matrices when constructing them, in the set of constructed matrices there are huge number of matrices are isomorphic, which means any matrix in the set

with column permutation is another matrix in the set. (I did not prove this, I am thinking because I enumberated all possibility in each reaction vector which means no matter how to permutate the columns (species) in a matrix, after permutation the matrix always falls in the same set). Now the set is closure for column (species) permutation, so does the set of $N_{(m-1)\times n}$, then if we construct the set of $N_{m\times n}$ from $N_{(m-1)\times n}$, we just need to add one in three reaction vectors (because with any column permutation we always get an isomorphic graph, the number of nth reaction vector is 3:
$[1, -1, 0, 0, \ldots, 0]$, $[1, 1, -1, 0, \ldots, 0]$, $[1, -1, -1, 0, \ldots, 0]$, the position of 1 and -1 are not important), in this treatment, we can reduce the number of matrices from $\binom{r}{m}$ to $\binom{r}{m-1}\cdot 3$.

I am not sure about if this step is correct

c). Further, we could reduce the number to $\binom{r}{m-2}\cdot 43$ for $n = 5$, or $\binom{r}{m-2}\cdot 46$ for $n \geq 6$. 43 and 46 are numbers of the unique reaction patterns for two reactions between n species. I listed these reaction vectors manually, and implemented it in the code.

1. Mass conservation;

Based on the stoichiometric matrix $N_{m\times n}$, we can construct a vector of mass values $\boldsymbol{m}$, m_i is the mass value of species S_i. Then we have the equation $N\boldsymbol{m} = \mathbf{0}$, because in each reaction the mass of left (reactants) is equal to mass of right (products). We need to make sure $\boldsymbol{m}$ is strictly positive.

a). firstly we check the rank of N (or linearly dependent), if $Rank(N) < m$, then it is linearly dependent, otherwise reject the matrix.

b). then caculate the nullspace basis of N. Then if the ith element in all basis is 0 or negative then m_i is 0 or negative. (This is not clear, may need some prove. When Maple compute the nullspace basis, it always return basis with 1s in e_j which means in the solution space $\sum x_j e_j$, x_j must be strictly positive.)

Actually a) and b) can be combined: if N is linearly independent, NullSpace will be empty. We could just exclude by examine nullspace basis.

Reference: Gevorgyan, A., Poolman, M. G., & Fell, D. A. (2008). Detection of stoichiometric inconsistencies in biomolecular models. Bioinformatics, 24(19), 2245-2251.

2. Complex banlanced: each species has at least one in flow and one out flow (this is very easy to check);

Exclude all matrices with any species (column) has no reactions involved, or all outflow (negative) or all inflow (positive)

Then we need to:

4. check competition: there are at least one species has two -1 and there is another -1 in each of the according reactions;

a). Get the $N\_$ which only have the negative elements in N. In negative matrix, check the RowSum get indices I of -2 and check the ColumnSum get indices J of $Cs_i \leq -2$, if there are two indices $i, h \in I$ and one index $j \in J$ with which $N_{hj} = N_{ij} = -1$, $i = 1..m$, then there is competition (of course the two competitors should be different, $c_1 \neq c_2$).

5. check loop: take indices of competitors, do the network searching, find the loop from one to another and then from the other to this one.

this is fairly easy to understand, I use breadth-first search.

a). first check if there are any species have more than two outflows (negative) then check if there are indeed two reactions with two species interact with another species. Then find the index of competitors.

b). use breadth-first search to search routes between competitors.

c). Note that to complete the competition loop (from bipartite graph), route from one competitor to the other need to avoid the competition reaction.

Cluster matrix into four categories: bistable with closed competition loop, bistable without closed competition loop, monostable with closed competition loop, monostable without closed competition loop.

All the procedure are implemented in the code. Any suggestions and corrections are more than welcome.

Initializations

```
> restart :
> interface(rtablesize = 400) :
> with(ListTools) :
> with(LinearAlgebra) :
> with(VectorCalculus) :
> with(GraphTheory) :
> with(combinat) :
> with(ArrayTools) :
> _Envsignum0 := 0 :
>
```

Functions for multistationality checking (execute before proceeding)

Functions for constructing stoichiomatric matrix and examine the existence of competition and closed loop.

1. To enumerate stoichiomatric vectors (reaction patterns)

List all reaction types based on the number of species. $R_n = \binom{n}{2} \cdot 2 + \binom{n}{3} \cdot 6$

```
> listRs := proc(n)
      local R, r, se, i, j, k, sign, l :
      r := binom(n, 2) · 2 + binom(n, 3) · 6 :
```

```
    R := Matrix(r, n) :
    # Now construct the reaction pattern matrix
    i := 1 :

    # here we first consider single transformation
    for se from -1 to 1 by 2 do
      for j from 1 to n by 1 do
        for k from j + 1 to n by 1 do
          R[i, j] := se :
          R[i, k] := -se :
          i := i + 1 :
        end do:
      end do:
    end do:

    # now consider heterodimerization and disassociation
    for sign from -1 to 1 by 2 do
      for j from 1 to n by 1 do
        for k from j + 1 to n by 1 do
          for l from k + 1 to n by 1 do
            R[i, j] := sign :
            R[i, k] := -sign :
            R[i, l] := -sign :
            i := i + 1 :
            R[i, j] := sign :
            R[i, k] := sign :
            R[i, l] := -sign :
            i := i + 1 :
            R[i, j] := sign :
            R[i, k] := -sign :
            R[i, l] := sign :
            i := i + 1 :
          end do:
        end do:
      end do:
    end do:

    # in this case we don't consider homodimerization and disassociation

    # Here transpose the R
    # R:=Transpose(R) :
     # no need to ranspose, need to assign rows to untransposed A's rows
    return(R) :
  end proc:
```

Here is a long function to enumerate first two reaction pattern (will reduce large mount of symmetric reactions).

```
> listR2 := proc(n)
    local R2, r2, R3, r3, R4, r4 :
    if n ≥ 4 then
      r2 := 29·2 :
      R2 := Matrix(r2, 4) :

      # in this case we don't consider homodimerization and disassociation

      ## now we construct the reaction pattern with n = 4
      R2[1] := ⟨1,-1⟩ : R2[2] := ⟨0, 0, 1,-1⟩ :
      R2[3] := ⟨1,-1⟩ : R2[4] := ⟨0, 1,-1⟩ :
      R2[5] := ⟨1,-1⟩ : R2[6] := ⟨1, 0,-1⟩ :
      R2[7] := ⟨1,-1⟩ : R2[8] := ⟨0,-1, 1⟩ :
      R2[9] := ⟨1,-1⟩ : R2[10] := ⟨-1, 0, 1⟩ :
      R2[11] := ⟨1,-1⟩ : R2[12] := ⟨-1, 1⟩ :
      R2[13] := ⟨1,-1⟩ : R2[14] := ⟨0, 1,-1,-1⟩ :
      R2[15] := ⟨1,-1⟩ : R2[16] := ⟨1, 0,-1,-1⟩ :
      R2[17] := ⟨1,-1⟩ : R2[18] := ⟨0,-1, 1,-1⟩ :
      R2[19] := ⟨1,-1⟩ : R2[20] := ⟨-1, 0, 1,-1⟩ :
      R2[21] := ⟨1,-1⟩ : R2[22] := ⟨0, 1, 1,-1⟩ :
      R2[23] := ⟨1,-1⟩ : R2[24] := ⟨1, 0, 1,-1⟩ :
      R2[25] := ⟨1,-1⟩ : R2[26] := ⟨0,-1, 1, 1⟩ :
      R2[27] := ⟨1,-1⟩ : R2[28] := ⟨-1, 0, 1, 1⟩ :
      R2[29] := ⟨1,-1⟩ : R2[30] := ⟨-1,-1, 1⟩ :
      R2[31] := ⟨1,-1⟩ : R2[32] := ⟨1, 1,-1⟩ :
      R2[33] := ⟨1,-1,-1⟩ : R2[34] := ⟨0, 1,-1,-1⟩ :
      R2[35] := ⟨1,-1,-1⟩ : R2[36] := ⟨1, 0,-1,-1⟩ :
      R2[37] := ⟨1,-1,-1⟩ : R2[38] := ⟨0,-1,-1, 1⟩ :
      R2[39] := ⟨1,-1,-1⟩ : R2[40] := ⟨-1,-1, 0, 1⟩ :
      R2[41] := ⟨1,-1,-1⟩ : R2[42] := ⟨0, 1, 1,-1⟩ :
      R2[43] := ⟨1,-1,-1⟩ : R2[44] := ⟨1, 1, 0,-1⟩ :
      R2[45] := ⟨1,-1,-1⟩ : R2[46] := ⟨0, 1,-1, 1⟩ :
      R2[47] := ⟨1,-1,-1⟩ : R2[48] := ⟨-1, 1, 0, 1⟩ :
      R2[49] := ⟨1,-1,-1⟩ : R2[50] := ⟨-1, 1, 1⟩ :
      R2[51] := ⟨1, 1,-1⟩ : R2[52] := ⟨0, 1, 1,-1⟩ :
      R2[53] := ⟨1, 1,-1⟩ : R2[54] := ⟨1, 1, 0,-1⟩ :
      R2[55] := ⟨1, 1,-1⟩ : R2[56] := ⟨0, 1,-1, 1⟩ :
      R2[57] := ⟨1, 1,-1⟩ : R2[58] := ⟨1,-1, 0, 1⟩ :


      if n ≥ 5 then
        ## now we add the reaction pattern with n = 5

        r3 := 43·2 :
        R3 := Matrix(r3, 5) :
        R3[1..r2] := R2[ ] :
```

```
    R3[59] := ⟨1,-1⟩ : R3[60] := ⟨0, 0, 1,-1,-1⟩ :
    R3[61] := ⟨1,-1⟩ : R3[62] := ⟨0, 0, 1, 1,-1⟩ :
    R3[63] := ⟨1,-1,-1⟩ : R3[64] := ⟨0, 0, 1,-1,-1⟩ :
    R3[65] := ⟨1,-1,-1⟩ : R3[66] := ⟨1, 0, 0,-1,-1⟩ :
    R3[67] := ⟨1,-1,-1⟩ : R3[68] := ⟨0, 0,-1, 1,-1⟩ :
    R3[69] := ⟨1,-1,-1⟩ : R3[70] := ⟨-1, 0, 0, 1,-1⟩ :
    R3[71] := ⟨1,-1,-1⟩ : R3[72] := ⟨0, 0, 1, 1,-1⟩ :
    R3[73] := ⟨1,-1,-1⟩ : R3[74] := ⟨1, 0, 0, 1,-1⟩ :
    R3[75] := ⟨1,-1,-1⟩ : R3[76] := ⟨0, 0,-1, 1, 1⟩ :
    R3[77] := ⟨1,-1,-1⟩ : R3[78] := ⟨-1, 0, 0, 1, 1⟩ :
    R3[79] := ⟨1, 1,-1⟩ : R3[80] := ⟨0, 0, 1, 1,-1⟩ :
    R3[81] := ⟨1, 1,-1⟩ : R3[82] := ⟨1, 0, 0, 1,-1⟩ :
    R3[83] := ⟨1, 1,-1⟩ : R3[84] := ⟨0, 0,-1, 1, 1⟩ :
    R3[85] := ⟨1, 1,-1⟩ : R3[86] := ⟨-1, 0, 0, 1, 1⟩ :

    if n ≥ 6 then
      ## now we add the reaction pattern with n = 6

      r4 := 46·2 :
      R4 := Matrix(r4, n) :
      R4[1..r3] := R3[ ] :

      R4[87] := ⟨1,-1,-1⟩ : R4[88] := ⟨0, 0, 0, 1,-1,-1⟩ :
      R4[89] := ⟨1,-1,-1⟩ : R4[90] := ⟨0, 0, 0, 1, 1,-1⟩ :
      R4[91] := ⟨1, 1,-1⟩ : R4[92] := ⟨0, 0, 0, 1, 1,-1⟩ :

      return(R4) :
    else
      return(R3) :
    end if:
  else
    return(R2) :
  end if:
else
  error "ERROR: n is smaller than 4"
end if:

end proc:
```

2. Now we can construct stoichiomatric matrix based on the reaction patterns, and examine their properties.

Here, we construct the stoichiomatric matrices.

The total number of stoichiomatric matrices is $\binom{R_n}{m}$, which is still a huge number. But currently there seems no other better options.

We only consider when $m \leq 6$.

The function(s) to examine existence of competition and loops a stoichiomatric matrix.

This function is used to check the existence of competition in the system

```
> existcompetition := proc(A)
    local i, j, m, n, count, check, An, Rs, Cs, Checks :
    An := (A - |A|)/2 :
    Rs := AddAlongDimension(A, 2) :
    Cs := AddAlongDimension(An, 1) :
    m := Dimension(A)[1] : n := Dimension(A)[2] :

    count := 0 :
    for j from 1 to n by 1 do
      if Cs[j] ≤ -2 then
        check := 0 :
        for i from 1 to m by 1 do
          if Rs[i] = -1 and A[i, j] ≤ -1 then
            check := check + 1 :
          end if:
        end do:

        if check ≥ 2 then
          count := count + 1 :
          return(count) :
        end if:
      end if:
    end do:
    return(count) :
  end proc:
```

This function is to check the existence of closed positive feedback loop with competition. It returns
a number if 0 then no competition, if 1 then only competition no loop, if 2 then with competition loop no switching (back), if 3 then with competition loop and switching (back).

```
> existcompetitionloop := proc(A)
    local i, j, m, n, count, check, An, Rs, Cs, Checks, exist, k, l, loops, switches, comps, p,
     q, v, r, s :
    An := (A - |A|)/2 :
    Rs := AddAlongDimension(A, 2) :
    Cs := AddAlongDimension(An, 1) :
    m := Dimension(A)[1] : n := Dimension(A)[2] :
```

```
count := 0 : exist := 0 : loops := 0 : switches := 0 :
#print(Rs);
#print(Cs);
for j from 1 to n by 1 do
  if Cs[j] ≤ -2 then
    check := 0 :
    Checks := Array( ) :
    k := 0 :
    for i from 1 to m by 1 do
      if Rs[i] = -1 and A[i,j] ≤ -1 then
        k := k + 1 :
        check := check + 1 :
        Checks(k) := i :
      end if:
    end do:
    #print(check);
    if check ≥ 2 then
      count := count + 1 :
      #print(count);
      comps := Size(Checks, 2) :
      v := Vector(5) :
      v[3] := j :
      for p from 1 to comps by 1 do
        v[1] := Checks(p) :
        if A[Checks[p], j] = -1 then
          for r from 1 to n by 1 do
            if A[Checks[p], r] = -1 and r ≠ j then
              v[4] := r :
            end if:
          end do:
          if v[4] = 0 then
            error "Can not find the other reactant" :
          end if:
        elif A[Checks[p], j] = -2 then
          v[4] := j :
        else
          error "The reactant is neither -1 nor -2" :
        end if:
        for q from p + 1 to comps by 1 do
          v[2] := Checks(q) :
          if A[Checks[q], j] = -1 then
            for r from 1 to n by 1 do
              if A[Checks[q], r] = -1 and r ≠ j then
                v[5] := r :
              end if:
            end do:
            if v[5] = 0 then
              error "Can not find the other reactant for second reaction" :
```

```
                    end if:
                elif A[Checks[q],j]=-2 then
                    v[5]:=j:
                else
                    error "The reactant is neither -1 nor -2 in second reaction":
                end if:
                #print(v);
                if v[4]≠v[5] then
                    l:=0:
                    l:=checkloop(A,v):
                    if l=1 then
                        loops:=loops+1:
                        #print(loops);
                    elif l=2 then
                        switches:=switches+1:
                        exist:=3:
                        return(exist):
                    end if:
                end if:
            end do:
        end do:
    end if:
end if:
end do:

if count≥1 then
    if loops≥1 then
        exist:=2:
    else
        exist:=1:
    end if:
end if:
return(exist):
end proc:
>
```

This function is used to check the competition loop (return 1) and switches (return 2), if no exist any of those return 0.

```
> checkloop:=proc(A,v)
    local exist,loop,switch,Q,i,j,k,m,n,visited:
    m:=Dimension(A)[1]:n:=Dimension(A)[2]:
    loop:=0:
    switch:=0:
    exist:=0:
    Q:=queue[new]():
    visited:=Vector(n):
    queue[enqueue](Q,v[4]):
```

```
while not queue[empty](Q) do
  j := queue[dequeue](Q) :
  for i from 1 to m do
    if i ≠ v[1] and i ≠ v[2] then
      if A[i,j] ≤ -1 then
        for k from 1 to n do
          if A[i, k] ≥ 1 then
            if k = v[5] then
              loop := loop + 1 :
              break:
            elif k ≠ v[4] then
              if visited[k] = 0 then
                queue[enqueue](Q, k) :
                visited[k] := 1 :
              end if:
            end if:
          end if:
        end do:
        if loop ≥ 1 then
          break:
        end if:
      end if:
    end if:
  end do:
  if loop ≥ 1 then
    queue[clear](Q) :
    break:
  end if:
end do:

queue[clear](Q) :
visited := Vector(n) :
if loop ≥ 1 then
  exist := 1 :
  for j from 1 to n do
    if j = v[4] then
      switch := switch + 1 :
      exist := 2 :
      return(exist) :
    elif A[v[2], j] ≥ 1 and visited[j] = 0 then
      queue[enqueue](Q, j) :
      visited[j] := 1 :
    end if:
  end do:

  while not queue[empty](Q) do
    j := queue[dequeue](Q) :
    for i from 1 to m do
```

```
            if i ≠ v[1] and i ≠ v[2] then
               if A[i,j] ≤ -1 then
                  for k from 1 to n do
                     if A[i, k] ≥ 1 then
                        if k = v[4] then
                           switch := switch + 1 :
                           exist := 2 :
                           return (exist) :
                        elif k ≠ v[5] and visited[k] = 0 then
                           queue[enqueue](Q, k) :
                           visited[k] := 1 :
                        end if:
                     end if:
                  end do:
               end if:
            end if:
         end do:
      end do:
   end if:

   loop := 0 :
   queue[clear](Q) :
   queue[enqueue](Q, v[5]) :
   visited := Vector(n) :
   while not queue[empty](Q) do
      j := queue[dequeue](Q) :
      for i from 1 to m do
         if i ≠ v[2] and i ≠ v[1] then
            if A[i,j] ≤ -1 then
               for k from 1 to n do
                  if A[i, k] ≥ 1 then
                     if k = v[4] then
                        loop := loop + 1 :
                        break:
                     elif k ≠ v[5] and visited[k] = 0 then
                        queue[enqueue](Q, k) :
                        visited[k] := 1 :
                     end if:
                  end if:
               end do:
               if loop ≥ 1 then
                  break:
               end if:
            end if:
         end if:
      end do:
      if loop ≥ 1 then
         queue[clear](Q) :
```

```
            break:
          end if:
        end do:

        queue[clear](Q) :
        visited := Vector(n) :
        if loop ≥ 1 then
          exist := 1 :
          for j from 1 to n do
            if j = v[5] then
              switch := switch + 1 :
              exist := 2 :
              return(exist) :
            elif A[v[1], j] ≥ 1 and visited[j] = 0 then
              queue[enqueue](Q, j) :
              visited[j] := 1 :
            end if:
          end do:

          while not queue[empty](Q) do
            j := queue[dequeue](Q) :
            for i from 1 to m do
              if i ≠ v[1] and i ≠ v[2] then
                if A[i, j] ≤ -1 then
                  for k from 1 to n do
                    if A[i, k] ≥ 1 then
                      if k = v[5] then
                        switch := switch + 1 :
                        exist := 2 :
                        return(exist) :
                      elif k ≠ v[4] and visited[k] = 0 then
                        queue[enqueue](Q, k) :
                        visited[k] := 1 :
                      end if:
                    end if:
                  end do:
                end if:
              end if:
            end do:
          end do:
        end if:

        return(exist) :
      end proc:
```

The function to check if the stoichiomatric matrix is mass conserved.

First check the passed matrix $A_{m\times n}$ (must be in a consistent form)

```
> ismassconserved_original := proc(A)
    local R, N, NS, m, n, absAdd, Add, x, y, z, e, i, nsAdd, nsAbsAdd, a, b, c :
    m := 0 :
    n := Dimension(A)[2] :
    absAdd := AddAlongDimension(|A|, 1) :
    z := Search(0, absAdd) :
    if z = 0 then
      Add := AddAlongDimension(A, 1) :
      x := Search(0, VectorAdd(absAdd, Add, 1, -1)) :
      if x = 0 then
        y := Search(0, VectorAdd(absAdd, Add, 1, 1)) :
        if y = 0 then
          N := NullSpace(A) :
          e := numelems(N) :
          if e > 0 then
            NS := Matrix(e, n) :
            for i from 1 to e by 1 do
              NS[i] := N[i] :
            end do:
            nsAbsAdd := AddAlongDimension(|NS|, 1) :
            a := Search(0, nsAbsAdd) :
            if a = 0 then
              nsAdd := AddAlongDimension(NS, 1) :
              b := Search(0, VectorAdd(nsAbsAdd, nsAdd, 1, 1)) :
              if b = 0 then
                m := 1 :
              end if:
            end if:
          end if:
        end if:
      end if:
    end if:
    return(m) :
  end proc:
```

The function to check if the stoichiomatric matrix is mass conserved.

First check the passed matrix $A_{m\times n}$ (must be in a consistent form)

```
> ismassconserved := proc(A)
    local R, N, NS, m, n, absAdd, Add, x, y, z, e, i, nsAdd, nsAbsAdd, a, b, c :
    m := 0 :
    n := Dimension(A)[2] :
    absAdd := AddAlongDimension(|A|, 1) :
    z := Search(0, absAdd) :
    if z = 0 then
      Add := AddAlongDimension(A, 1) :
      x := Search(0, VectorAdd(absAdd, Add, 1, -1)) :
      if x = 0 then
        y := Search(0, VectorAdd(absAdd, Add, 1, 1)) :
```

```
            if y = 0 then
               N := NullSpace(A) :
               e := numelems(N) :
               if e > 0 then
                  NS := Matrix(e, n) :
                  for i from 1 to e by 1 do
                     NS[i] := N[i] :
                  end do:
                  nsAbsAdd := AddAlongDimension(|NS|, 1) :
                  a := Search(0, nsAbsAdd) :
                  if a = 0 then
                     nsAdd := AddAlongDimension(NS, 1) :
                     b := Search(0, VectorAdd(nsAbsAdd, nsAdd, 1, 1)) :
                     if b = 0 then
                        m := 1 :
                        # here implement sufficient check of mass conservation

                     end if:
                  end if:
               end if:
            end if:
         end if:
         return(m) :
      end proc:
```

To get the indecies of vectors in a Matrix

```
> indicesInMatrix := proc(B, A)
      ## return the indices of vectors from B in Matrix A
      local n, m, p, q, Ind, i, j :
      p := Dimension(A)[1] :
      q := Dimension(A)[2] :
      m := Dimension(B)[1] :
      n := Dimension(B)[2] :

      if n ≠ q then
         error "ERROR: The two matrices have the different column number."
      end if:

      Ind := Vector(m) :
      for i from 1 to m by 1 do
         for j from 1 to p by 1 do
            if IsEqual(B[i], A[j]) then
               Ind[i] := j :
               break:
            end if:
```

```
      end do:
      if Ind[i] = 0 then
        error "ERROR: no index found for ith vector from B in A"
      end if:
    end do:

    return (Ind) :
  end proc:
```

Construct and examine the properties of all stoichiomatric matrices.

```
> constrM := proc(n, m)
    local R, tA, A, iA, Z, r, g, h, i, j, k, l, total, right, mc, V, R2, r2, injective, injective0,
     injective1, injectiveEx, fileName, matrixData, interV, comp, injectiveEx0,
     injectiveEx1, injectiveEx2, injectiveEx3, s, selected, pfloops, unique, pfintersect,
     pfcount, f, myset, Ind :
    r := binomial(n, 2)·2 + binomial(n, 3)·6 :

    A := Matrix(m, n) :

    R := listRs(n) :
    R2 := listR2(n) :
    Ind := indicesInMatrix(R2, R) :

    if n = 4  then  r2 := 29·2 : end if:
    if n = 5  then  r2 := 43·2 : end if:
    if n ≥ 6  then  r2 := 46·2 : end if:
    if n < 4 then error "ERROR: n is smaller than 4" end if:

    total := r2/2 · binomial(r, m − 2) :

    # here we use some variable to count how many reactions are correct.
    right := 0 : injective0 := 0 : injective1 := 0 :
    #injectiveEx0:= 0:injectiveEx1:= 0:injectiveEx2:= 0: injectiveEx3:= 0:
    for l from 1 to r2 − 1 by 2 do

      myset := {seq(i, i = 1 ..m − 2)} :

      A[1] := R2[l] :
      A[2] := R2[l + 1] :
      while myset ≠ FAIL do
        if Ind[l] in myset or Ind[l + 1] in myset then
          myset := nextcomb(myset, r) :
        else
```

```
        A[3..m] := R[[op(myset)]]:
        mc := ismassconserved(A):
        if mc = 1 then
          right := right + 1:

          # check the existence of compeition and competition loop.
          comp := existcompetitionloop(A):

          tA := Transpose(A):

          # check the existence of intersecting positive feedback loops
          Z := findZ(tA): s := Rank(tA): selected := findloops(tA, Z):
          pfloops := numelems(selected):
          pfintersect := 0:
          if pfloops ≥ 2 then
            pfcount := 0:
            for f from 1 to numelems(selected) by 1 do
              pfcount := pfcount + numelems(selected[f]):
            end do:
            unique := []:
            for f from 1 to numelems(selected) by 1 do
              unique := [op(unique), op(selected[f])]:
            end do:
            if pfcount > numelems(MakeUnique(unique)) then
              pfintersect := 1:
            end if:
          end if:

          iA := Transpose(A):
          # simple injectivity check
          injective := isinjective(iA):
          if injective = 0 then
            injective0 := injective0 + 1:
            injectiveEx := isinjectiveextended(iA):
            if injectiveEx = 1 or injectiveEx = 3 then
              if comp = 3 then
                if pfintersect = 1 then
                  fileName
 := sprintf(
"%1dspecies/nonmultistationary/competitionloop_intersectingloops/injectiveEx%1d_\
%d.csv", n, injectiveEx, right):
                  ExportMatrix(fileName, iA, target = csv, format = rectangular,
mode = ascii):
                else
                  fileName
 := sprintf(
"%1dspecies/nonmultistationary/competitionloop_nointersectingloops/injectiveEx%1\
d_%d.csv", n, injectiveEx, right):
```

```
                    ExportMatrix(fileName, iA, target = csv, format = rectangular,
mode = ascii) :
                end if:
              else
                if pfintersect = 1 then
                  fileName
 := sprintf(
"%1dspecies/nonmultistationary/nocompetitionloop_intersectingloops/injectiveEx%1\
d_%d.csv", n, injectiveEx, right) :
                    ExportMatrix(fileName, iA, target = csv, format = rectangular,
mode = ascii) :
                else
                  fileName
 := sprintf(
"%1dspecies/nonmultistationary/nocompetitionloop_nointersectingloops/injectiveEx\
%1d_%d.csv", n, injectiveEx, right) :
                    ExportMatrix(fileName, iA, target = csv, format = rectangular,
mode = ascii) :
                end if:
              end if:
            elif injectiveEx = 0 or injectiveEx = 2 then
              if comp = 3 then
                if pfintersect = 1 then
                  fileName
 := sprintf(
"%1dspecies/multistationary/competitionloop_intersectingloops/injectiveEx%1d_%d.
csv", n, injectiveEx, right) :
                    ExportMatrix(fileName, iA, target = csv, format = rectangular,
mode = ascii) :
                else
                  fileName
 := sprintf(
"%1dspecies/multistationary/competitionloop_nointersectingloops/injectiveEx%1d_%\
d.csv", n, injectiveEx, right) :
                    ExportMatrix(fileName, iA, target = csv, format = rectangular,
mode = ascii) :
                end if:
              else
                if pfintersect = 1 then
                  fileName
 := sprintf(
"%1dspecies/multistationary/nocompetitionloop_intersectingloops/injectiveEx%1d_%\
d.csv", n, injectiveEx, right) :
                    ExportMatrix(fileName, iA, target = csv, format = rectangular,
mode = ascii) :
                else
                  fileName
 := sprintf(
```

```
"%1dspecies/multistationary/nocompetitionloop_nointersectingloops/injectiveEx%1d\
_%d.csv", n, injectiveEx, right) :
                    ExportMatrix(fileName, iA, target = csv, format = rectangular,
mode = ascii) :
                  end if:
                end if:
              else
                error "ERROR: injectivity extended of A is not any of 0 to 3."
              end if:
            elif injective = 1 then
              injective1 := injective1 + 1 :
              if comp = 3 then
                if pfintersect = 1 then
                  fileName
 := sprintf(
"%1dspecies/nonmultistationary/competitionloop_intersectingloops/injective%1d_%d\
.csv", n, injective, right) :
                  ExportMatrix(fileName, iA, target = csv, format = rectangular,
mode = ascii) :
                else
                  fileName
 := sprintf(
"%1dspecies/nonmultistationary/competitionloop_nointersectingloops/injective%1d_\
%d.csv", n, injective, right) :
                  ExportMatrix(fileName, iA, target = csv, format = rectangular,
mode = ascii) :
                end if:
              else
                if pfintersect = 1 then
                  fileName
 := sprintf(
"%1dspecies/nonmultistationary/nocompetitionloop_intersectingloops/injective%1d_\
%d.csv", n, injective, right) :
                  ExportMatrix(fileName, iA, target = csv, format = rectangular,
mode = ascii) :
                else
                  fileName
 := sprintf(
"%1dspecies/nonmultistationary/nocompetitionloop_nointersectingloops/injective%1\
d_%d.csv", n, injective, right) :
                  ExportMatrix(fileName, iA, target = csv, format = rectangular,
mode = ascii) :
                end if:
              end if:
            else
              error "ERROR: the injectivity of iA is neither 0 nor 1."
            end if:
          end if:
```

```
                myset := nextcomb(myset, r) :
            end if:
        end do:
    end do:

    V := [injective0, injective1, right, total, r, r2] :
    return(V) :
end proc:
>
```

```
> V := constrM(5, 5)
    # (original algorithm) check competition loop and intersecting loops ~ (36s to )
                        V := [4707, 4522, 9229, 3532880, 80, 86]                    (1)
> V := constrM(5, 5) # new algorithm to exclude potentially duplicated reactions in matrix
                        V := [4593, 4340, 8933, 3532880, 80, 86]                    (2)
> V := constrM(4, 4) # 4 species and 4 reactions
                          V := [70, 166, 236, 18270, 36, 58]                        (3)
> V := constrM(4, 5) # 4 species and 5 reactions
                         V := [560, 728, 1288, 207060, 36, 58]                      (4)
> V := constrM(4, 6) # 4 species and 6 reactions
                        V := [1773, 1384, 3157, 1708245, 36, 58]                    (5)
> V := constrM(5, 3) # 5 species and 3 reactions
                          V := [0, 0, 0, 3440, 80, 86]                              (6)
> V := constrM(5, 4) # 5 species and 4 reactions
                         V := [57, 91, 148, 135880, 80, 86]                         (7)
```

▶ Testing